
Diese Übersicht fasst die im Buch verwendeten DAX-Muster kompakt zusammen. Sie ist als Nachschlageblatt gedacht 
und verwendet dieselben Tabellen- und Measure-Namen wie der Praxisfall. Die Formeln sind in der Buchschreibweise 
mit Semikolon als Argumenttrennzeichen dargestellt.

Voraussetzung: Das Datenmodell ist vollständig aufgebaut, DimKalender ist als Datumstabelle markiert und alle 14 
Beziehungen sind aktiv.

1. Kontroll- und Mengenkennzahlen

Anzahl Belegzeilen =
COUNTROWS(FactUmsatz)

Anzahl Rechnungen =
DISTINCTCOUNT(FactUmsatz[Rechnungsnummer])

Anzahl Stornozeilen =
CALCULATE(
    COUNTROWS(FactUmsatz);
    FactUmsatz[Buchungsstatus] = "Storno"
)

Absatzmenge =
SUM(FactUmsatz[Menge])

Kennzahl Gesamtwert

Anzahl Belegzeilen 4.092

Anzahl Rechnungen 2.918

Anzahl Stornozeilen 91

Absatzmenge 20.319

2. Umsatz, Wareneinsatz und Deckungsbeitrag

Umsatz =
SUMX(
    FactUmsatz;
    FactUmsatz[Menge]
        * FactUmsatz[Einzelpreis_EUR]
        * (1 - FactUmsatz[Rabatt_Prozent])
)

Wareneinsatz =
SUMX(
    FactUmsatz;
    FactUmsatz[Menge]
        * RELATED(DimProdukt[Standardkosten_EUR])
)

Deckungsbeitrag =
[Umsatz] - [Wareneinsatz]

Deckungsbeitragsquote =
DIVIDE(
    [Deckungsbeitrag];
    [Umsatz]
)

Kennzahl Gesamtwert

Umsatz 6.561.846,32 €

Wareneinsatz 4.331.529,00 €

Deckungsbeitrag 2.230.317,32 €

Deckungsbeitragsquote 33,99 %

3. Plan-Ist-Kennzahlen

Umsatz Plan =
SUM(FactBudget[Umsatz_Plan_EUR])

Deckungsbeitrag Plan =
SUM(FactBudget[Deckungsbeitrag_Plan_EUR])

Umsatzabweichung =
[Umsatz] - [Umsatz Plan]

Umsatzabweichung % =
DIVIDE(
    [Umsatzabweichung];
    [Umsatz Plan]
)

Planerreichung =
DIVIDE(
    [Umsatz];
    [Umsatz Plan]
)

Kennzahl Gesamtwert

Umsatz Plan 12.472.158,00 €

Umsatzabweichung -5.910.311,68 €

Umsatzabweichung % -47,39 %

Planerreichung 52,61 %

4. Kosten und Betriebsergebnis

Personalkosten =
SUM(FactPersonalkosten[Personalkosten_Gesamt_EUR])

Gemeinkosten =
SUM(FactGemeinkosten[Betrag_EUR])

Betriebsergebnis =
[Deckungsbeitrag]
    - [Personalkosten]
    - [Gemeinkosten]

Kennzahl Gesamtwert

Personalkosten 2.242.285,15 €

Gemeinkosten 944.043,13 €

Betriebsergebnis -956.010,96 €

Fachliche Einordnung: Das Betriebsergebnis ist im Lernfall bewusst vereinfacht. Abschreibungen, Finanzergebnis, 
Steuern und weitere GuV-Positionen sind nicht enthalten.

5. Filterkontext mit CALCULATE
CALCULATE  wertet  ein  Measure  in  einem  veränderten  Filterkontext  aus.  Dadurch  lassen  sich  feste  Teilmengen,  
Gesamtanteile oder Vergleichswerte ausdrücken, ohne die Basiskennzahl zu duplizieren.

Umsatz Hardware =
CALCULATE(
    [Umsatz];
    DimProduktgruppe[Produktgruppe] = "Hardware"
)

Umsatz alle Regionen =
CALCULATE(
    [Umsatz];
    REMOVEFILTERS(DimKunde[Region])
)

Umsatzanteil Region =
DIVIDE(
    [Umsatz];
    [Umsatz alle Regionen]
)

Kontrollwert: Umsatz Hardware = 3.928.253,52 €.

6. Zeitintelligenz

Umsatz Vorjahr =
CALCULATE(
    [Umsatz];
    SAMEPERIODLASTYEAR(DimKalender[Datum])
)

Umsatz Vormonat =
CALCULATE(
    [Umsatz];
    DATEADD(
        DimKalender[Datum];
        -1;
        MONTH
    )
)

Umsatz YTD =
TOTALYTD(
    [Umsatz];
    DimKalender[Datum]
)

6.1 Rollierende zwölf Monate

Umsatz rollierend 12 Monate =
VAR LetztesDatum =
    MAX(DimKalender[Datum])
RETURN
    CALCULATE(
        [Umsatz];
        DATESINPERIOD(
            DimKalender[Datum];
            LetztesDatum;
            -12;
            MONTH
        )
    )

Prüfung im Kontext Juni 2026 Erwarteter Wert

Umsatz Juni 2026 360.723,11 €

Umsatz Juni 2025 353.898,60 €

Umsatz Mai 2026 363.964,60 €

Umsatz YTD 2026 2.194.792,41 €

Umsatz YTD Vorjahr 2.207.184,14 €

YTD-Abweichung -12.391,74 € / -0,56 %

Rollierend 12 Monate 4.354.662,17 €

Zeitregel: Zeitmeasures verwenden DimKalender[Datum]. Für Monatsnamen muss die Sortierung nach 
DimKalender[Monat_Nr] eingestellt sein.

7. DIVIDE, VAR und robuste Schreibweise
DIVIDE ist für Quoten vorzuziehen, weil ein Nenner von null oder BLANK kontrolliert behandelt wird. VAR verbessert  
die Lesbarkeit, wenn ein Zwischenergebnis mehrfach benötigt wird oder eine Berechnung in logische Schritte zerlegt 
werden soll.

Beispielquote =
DIVIDE(
    [Zähler];
    [Nenner]
)

Beispiel mit VAR =
VAR WertA = [Umsatz]
VAR WertB = [Umsatz Plan]
RETURN
    WertA - WertB

8. Formatierung und Fehlersuche
Typ Empfohlenes Format

Anzahl / Menge Ganze Zahl, 0 Dezimalstellen

Geldbeträge Währung Euro, 2 Dezimalstellen

Quoten / Abweichung % Prozent, 1-2 Dezimalstellen

Durchschnittspreise Währung Euro, 2 Dezimalstellen

Fehlerbild Zuerst prüfen

Umsatz weicht ab Power-Query-Bereinigung, Rabattwerte, Dubletten

Wareneinsatz ist leer Beziehung DimProdukt -> FactUmsatz und RELATED

Budget reagiert nicht auf Produktgruppe DimProduktgruppe und Beziehungen zu DimProdukt / FactBudget

Vorjahreswert ist leer Kalenderzeitraum und aktive Datumsbeziehung

Monate alphabetisch Monatsname nach Monat_Nr sortieren

Rollierende 12 Monate auf Karte unerwartet Letztes sichtbares Datum / Filter auf Datenmonat

9. Endkontrolle der Grundkennzahlen
Kennzahl Sollwert

Anzahl Belegzeilen 4.092

Anzahl Rechnungen 2.918

Anzahl Stornozeilen 91

Absatzmenge 20.319

Umsatz 6.561.846,32 €

Wareneinsatz 4.331.529,00 €

Deckungsbeitrag 2.230.317,32 €

Deckungsbeitragsquote 33,99 %

Umsatz Plan 12.472.158,00 €

Planerreichung 52,61 %

Personalkosten 2.242.285,15 €

Gemeinkosten 944.043,13 €

Betriebsergebnis -956.010,96 €

Freigaberegel: Kein Berichtslayout wird weitergebaut, solange ein verbindlicher Kontrollwert ungeklärt abweicht. 
Zuerst Quelle, Power Query, Beziehungen und Datentypen prüfen.

